

Zdzisław Kes

Uniwersytet Ekonomiczny we Wrocławiu

ZAAWANSOWANE FORMUŁY OBLICZENIOWE MS EXCELA W ROZWIĄZANIU PROBLEMU KALKULACJI CEN OBSŁUGI LOGISTYCZNEJ

Streszczenie: Artykuł zawiera opis problematyki zastosowania arkusza kalkulacyjnego w controllingu logistyki. Informatyczne wspomaganie w tym zakresie jest bardzo ważnym elementem systemu informacyjnego. W sytuacji, gdy system wykorzystuje głównie arkusze kalkulacyjne, ważne jest zastosowanie zaawansowanych funkcji i formuł sprzyjających efektywnemu wykorzystaniu narzędzia. Do tych funkcji autor zaliczył formuły tablicowe.

Słowa kluczowe: MS Excel, informatyczne wspomaganie controllingu, controlling logistyki, formuły tablicowe.

1. Wstęp

Kanwą do napisania tego artykułu jest praktyczny problem występujący podczas alokacji zasobów na działania. Chodzi o taką sytuację, w której poszczególne zasoby, będące w posiadaniu przedsiębiorstwa, są przypisane do realizowanych przez nie działań ilościowo lub wartościowo w postaci normatywów lub cenników. Natomiast działania są wykonywane według harmonogramu w sposób powtarzalny. Problem z kalkulacją następuje wówczas, gdy trzeba obliczyć sumy iloczynów stawek poszczególnych zasobów i ilości planowanych bądź zrealizowanych działań. Mimo że brzmi to trywialnie, to implementacja takich obliczeń w warunkach korzystania np. z MS Excela może nastężyć wielu interesujących dylematów.

W niniejszym artykule zostaną opisane i poddane krytyce różne podejścia implementacji obliczeń do arkuszy kalkulacyjnych. Zostaną zaprezentowane typowe funkcje MS Excela, funkcje zaawansowane wykorzystujące formuły macierzowe oraz narzędzia obróbki danych mające zastosowanie w wykonywaniu omawianych kalkulacji.

Przy budowie przykładu prezentującego omawiane zagadnienia przyjęto założenia, że zasobami są rodzaje pojazdów będących w dyspozycji przedsiębiorstwa transportowego, świadczącego kompleksowe usługi logistyczne w postaci przewozu zestandaryzowanych towarów na kilku trasach (trasa stanowi odpowiednik działania). Wykorzystanie danego rodzaju pojazdu na określonej trasie wiąże się z określo-

nymi w cenniku kosztami. W trakcie roku przedsiębiorstwo świadczy przewozy na tych trasach różną liczbę razy. W przykładzie chodzi o przedstawienie różnych sposobów obliczeń kosztów tras w przekroju używanych pojazdów.

2. Założenia wstępne

Dane wykorzystywane do obliczeń składają się z dwóch zestawów. Pierwszy zestaw obejmuje cennik wykorzystywania różnych pojazdów na poszczególnych trasach (tabela 1). Dla ułatwienia prezentacji zagadnień tworzenia formuł założono, że dwa rodzaje pojazdów obsługuje pięć standardowych tras.

Tabela 1. Cennik

Trasa	Rodzaj obsługujących pojazdów	Cena [zł]
Trasa_1	Pojazd_1	200
Trasa_1	Pojazd_2	150
Trasa_2	Pojazd_1	300
Trasa_2	Pojazd_2	200
Trasa_3	Pojazd_1	400
Trasa_3	Pojazd_2	300
Trasa_4	Pojazd_1	100
Trasa_4	Pojazd_2	50
Trasa_5	Pojazd_1	500
Trasa_5	Pojazd_2	400

Źródło: dane przykładowe.

Drugi zestaw danych obejmuje harmonogram świadczenia usług na wymienionych trasach w poszczególnych miesiącach (tabela 2). Harmonogram zawiera liczbę wyjazdów grupy pojazdów, która obsługuje daną trasę.

Tabela 2. Harmonogram usług

Miesiące	1	2	3	4	5	6	7	8	9	10	11	12
Trasa_1	100	50	100	100	100	100	100	100	40	120	130	305
Trasa_2	10	150	250	250	250	250	250	250	70	230	165	340
Trasa_3	10	250	400	500	1000	1000	1000	500	100	25	200	375
Trasa_4	20	350	550	40	40	60	30	30	30	60	235	410
Trasa_5	20	200	30	30	50	70	10	10	10	95	270	445

Źródło: dane przykładowe.

W rezultacie obliczeń powinna zostać podana łączna kwota kosztów związanych ze świadczeniem usług w danym miesiącu wraz z podaniem obciążeń generowanych przez poszczególne rodzaje pojazdów. Jest to istotne z punktu widzenia rozliczeń kosztów i przy obliczaniu rentowności poszczególnych rodzajów środków transportowych. Wyniki należy podać według układu zaprezentowanego w tabeli 3.

Tabela 3. Wyniki obliczeń – wartość wykonanych usług

Miesiące	1	2	3	4	5	6	7	8	9	10	11	12
Pojazd_1												
Pojazd_2												
Razem												
Udział %												

Źródło: opracowanie własne.

Jak wcześniej zauważono, przeprowadzenie obliczeń wymaga przemnożenia ceny dla rodzaju pojazdu, ustalonej w ramach danej trasy, przez ustaloną liczbę wyjazdów na tej trasie w konkretnym miesiącu. W związku z tym, że obsługiwanych jest wiele tras w danym miesiącu, mamy do czynienia z sumą iloczynów cen i liczby wyjazdów w ramach wszystkich tras.

Tak prosty problem może jednak napotkać w warunkach przeprowadzania obliczeń w arkuszu kalkulacyjnym MS Excel bardzo ciekawe implikacje. Problematyczne staje się zastosowanie formuł, które będą mogły przemnożyć dwa czynniki wyznaczone przez rodzaj pojazdu oraz numer trasy. Zbudowanie takiej formuły w przypadku wielu grup pojazdów i dużej liczebności różnych tras może spowodować bardzo dużo komplikacji. W kolejnych podpunktach zostaną zaprezentowane i omówione różne sposoby rozwiązywania tego zagadnienia. W pierwszym podejściu zostanie pokazana specyficzna właściwość funkcji MS Excela SUMA.ILOCZYNÓW. W kolejnym punkcie zostaną wykorzystane formuły macierzowe (tablicowe). Trzeci sposób rozwiązania zakłada konieczność niewielkiej modyfikacji danych wyjściowych, co pozwala w rezultacie na zastosowanie łatwej metody obliczeń. W ostatnim podpunkcie zostaną zaprezentowane dodatkowe rozwiązania związane z modyfikacją danych wejściowych, pozwalającą na znaczne uproszczenia formuł obliczeniowych.

3. Specyficzne zastosowanie funkcji SUMA.ILOCZYNÓW

Jak wcześniej wskazywano, obliczenie wartości usług świadczonych przez dany rodzaj pojazdu wymaga obliczenia sumy iloczynów dwóch czynników: cen i ilości. Wyznaczając tę sumę w pierwszym miesiącu dla pojazdu oznaczonego jako pojazd_1, należy dokonać następujących obliczeń dla każdej obsługiwanej trasy (w pierwszym miesiącu obsługiwano ich pięć, czyli wszystkie trasy):

1. Dla trasy 1 cena wyznaczona dla pojazdu_1 wynosi 200 zł/wyjazd.
2. Liczba wyjazdów wynosi 100 razy.
3. Iloczyn ceny i liczby wyjazdów wynosi 20 000 zł.

W celu obliczenia wspomnianego iloczynu należy zbudować formułę, która uwzględni właściwą wartość z tabeli *Ceny* na podstawie nazwy rodzaju pojazdu i obsługiwanej przez niego trasy. Oznacza to wybór z zakresu komórek z cenami tej, która odpowiada rodzajowi pojazdu i numerowi trasy. Technicznie mamy do czynienia z koniunkcją dwóch kryteriów i dokonaniem na tej podstawie wyboru. W celu ułatwienia prezentacji rozwiązania posłużymy się odwołaniami do konkretnych komórek arkusza kalkulacyjnego, który we fragmencie został pokazany na rysunku 1.

	A	B	C	D
1	CENNIK			
2	Trasa	Rodzaj obsługujących pojazdów	Cena	
3	Trasa_1	Pojazd_1	200	
4	Trasa_1	Pojazd_2	150	
5	Trasa_2	Pojazd_1	300	
6	Trasa_2	Pojazd_2	200	
7	Trasa_3	Pojazd_1	400	
8	Trasa_3	Pojazd_2	300	
9	Trasa_4	Pojazd_1	100	
10	Trasa_4	Pojazd_2	50	
11	Trasa_5	Pojazd_1	500	
12	Trasa_5	Pojazd_2	400	
13				
14	HARMONOGRAM WYJAZDÓW			
15	Miesiące	1	2	3
16	Trasa_1	100	50	100
17	Trasa_2	10	150	250
18	Trasa_3	10	250	400
19	Trasa_4	20	350	550
20	Trasa_5	20	200	30
21				
22	SUMA WARTOŚCI			
23	Miesiące	1	2	3
24	Pojazd_1			
25	Pojazd_2			
26	Razem			

Rys. 1. Rozwiązanie nr 1

Źródło: fragment okna programu MS Excel.

Kryteria wyboru wartości z tabeli *Cennik* (dla komórki B24) można zapisać w następujący sposób:

$$B3:B12=A24 \text{ oraz } A3:A12=A16,$$

gdzie: B3:B12 – zakres komórek tabeli cennika z rodzajami pojazdów,
 A24 – komórka z oznaczeniem rodzaju pojazdu, dla którego wykonuje się obliczenia,
 A3:A12 – zakres komórek tabeli cennika z numerami tras,
 A16 – komórka z oznaczeniem numeru trasy, dla której wykonuje się obliczenia.

Zastosowanie tego podwójnego kryterium zwraca komórkę C3 (z wartością 200) oznaczającą cenę za wykorzystanie pojazdu_1 na trasie_1. Jeżeli wartość z komórki C3 zostanie pomnożona przez liczbę wyjazdów na trasie_1 (komórka B16) uzyskujemy rozwiązanie pierwszej iteracji. Powtarzając tę procedurę tyle razy, ile jest tras, otrzymamy całkowitą wartość wykorzystywanego środka transportowego w danym miesiącu.

Niestety, w arkuszu kalkulacyjnym do wersji 2003 nie istnieją funkcje pozwalające na dokonywanie wielokryterialnego wyboru wartości z zakresu¹. Pewnym rozwiązaniem tego braku jest właśnie specyficzne zastosowanie funkcji SUMA.ILOCZYNÓW. Specyficzność polega na zadeklarowaniu argumentów oraz wartości, która ma zostać zwrócona. Postać tej funkcji przedstawiono w tabeli 4.

Tabela 4. Funkcja SUMA.ILOCZYNÓW

Adres formuły	Formuła
B24	=SUMA.ILOCZYNÓW((\$B\$3:\$B\$12=\$A24)*(\$A\$3:\$A\$12=\$A\$16);\$C\$3:\$C\$12)
Kroki	Wyniki działania szacowania funkcji
\$B\$3:\$B\$12=\$A24	{Prawda/Falsz/Prawda/Falsz/Prawda/Falsz/Prawda/Falsz/Prawda/Falsz}
\$A\$3:\$A\$12=\$A\$16	{Prawda/Prawda/Falsz/Falsz/Falsz/Falsz/Falsz/Falsz/Falsz/Falsz}
(\$B\$3:\$B\$12=\$A24)* (\$A\$3:\$A\$12=\$A\$16)	{1/0/0/0/0/0/0/0/0/0}
\$C\$3:\$C\$12	{200/150/300/200/400/300/100/50/500/400}
SUMA.ILOCZYNÓW(...)	200

Źródło: opracowanie własne.

Jak pokazano w tabeli 4, funkcja SUMA.ILOCZYNÓW na podstawie koniunkcji dwóch kryteriów jest w stanie wybrać z tabeli z cennikiem właściwą wartość odpowiadającą danemu rodzajowi pojazdu obsługującego wskazaną trasę. Oznacza to, że omawiana funkcja, zamiast funkcji matematycznej, staje się funkcją wyszuki-

¹ W wersji Excela 2007 do tego typu obliczeń dodano funkcję SUMA.WARUNKÓW.

wania, co świadczy o jej dużym potencjale. Dodatkowo należy zwrócić uwagę na zastosowanie, jako argumentu tej funkcji, iloczynu dwóch równań z zakresów komórek. Równania te w sposób niejawni dokonują selekcji argumentów (rodzaju pojazdu oraz nr trasy) z zakresów komórek znajdujących się w tabeli *Cennik*. W wyniku selekcji otrzymujemy dwie macierze z wartościami logicznymi, które dopiero po przemnożeniu (chodzi o iloczyn logiczny) dają macierz z zerami lub jedynkami. Dopiero tak skonwertowana macierz może być podstawą do dokonania typowych obliczeń w ramach funkcji SUMA.ILOCZYNÓW, które w rezultacie dają wynik wskazujący na wartość wyznaczoną przez rodzaj pojazdu i numer trasy. Wynik funkcji SUMA.ILOCZYNÓW należy następnie przemnożyć przez liczbę wyjazdów na danej trasie i powtórzyć te obliczenia w podobny sposób dla pozostałych tras. Dopiero suma tych obliczeń będzie właściwą odpowiedzią na pytanie, jaka jest wartość wyjazdów danego rodzaju pojazdu w danym miesiącu.

Takie podejście do rozwiązania niestety niesie niebezpieczeństwo popełnienia błędu we wprowadzaniu skomplikowanej formuły oraz konieczność jej znacznej modyfikacji przy każdej zmianie warunków wyjściowych w postaci liczby obsługiwanych tras. Dlatego w dalszej części opracowania zostaną podane modyfikacje tego rozwiązania.

4. Specyficzne zastosowanie funkcji SUMA

Modyfikacją rozwiązania z punktu poprzedniego jest zastosowanie specyficzne funkcji SUMA. Specjalne podejście do tej funkcji polega również na wykorzystaniu jej jako funkcji wyszukiwującej, a nie matematycznej. Jest to możliwe dzięki wykorzystaniu formuł macierzowych w połączeniu z funkcjami arkuszowymi. To połączenie pozwala na zbudowanie za pomocą funkcji SUMA bardzo zaawansowanych formuł wyszukiwujących określone wartości z zakresu na podstawie kilku bądź kilkunastu kryteriów zapisanych w podobnym formacie jak we wcześniejszym przypadku. Dla ułatwienia rozważań zostanie zaprezentowana formuła z komórki B24 pokazanej na rysunku 2.

Aby uprościć wprowadzanie formuł, w rozwiązaniu nr 2 zastosowano dwie zmiany w stosunku do rozwiązania nr 1. Po pierwsze w tabeli *Cennik* wprowadzono kolumnę *Indeks* będącą złożeniem dwóch innych kolumn: trasa i rodzaj obsługujących pojazdów. Kolumna ta powstała poprzez zastosowanie w niej formuły z operatorem ampersand (&) wykorzystywanym do łączenia wartości o różnych formatach. W tym przypadku połączono wartość z kolumny *Trasa* z wartością kolumny *Rodzaj obsługujących pojazdów*, otrzymując w ten sposób kolumnę *Indeks* (nagłówek tej kolumny znajduje się w komórce C2 na rysunku 2). Druga modyfikacja dotyczy rozbudowania tabeli z wynikami obliczeń (*Suma wartości*), gdzie dodano po pięć wierszy poniżej każdego wiersza zawierającego sumę kosztów eksploatacji pojaz-

1	2	A	B	C	D	E
	1	CENNIK				
	2	Trasa	Rodzaj obsługujących pojazdów	Indeks	Cena	
	3	Trasa_1	Pojazd_1	Trasa_1Pojazd_1	200	
	4	Trasa_1	Pojazd_2	Trasa_1Pojazd_2	150	
	5	Trasa_2	Pojazd_1	Trasa_2Pojazd_1	300	
	6	Trasa_2	Pojazd_2	Trasa_2Pojazd_2	200	
	7	Trasa_3	Pojazd_1	Trasa_3Pojazd_1	400	
	8	Trasa_3	Pojazd_2	Trasa_3Pojazd_2	300	
	9	Trasa_4	Pojazd_1	Trasa_4Pojazd_1	100	
	10	Trasa_4	Pojazd_2	Trasa_4Pojazd_2	50	
	11	Trasa_5	Pojazd_1	Trasa_5Pojazd_1	500	
	12	Trasa_5	Pojazd_2	Trasa_5Pojazd_2	400	
	13					
	14	HARMONOGRAM WYJAZDÓW				
	15	Miesiące	1	2	3	4
	16	Trasa_1	100	50	100	
	17	Trasa_2	10	150	250	
	18	Trasa_3	10	250	400	
	19	Trasa_4	20	350	550	
	20	Trasa_5	20	200	30	
	21					
	22	SUMA WARTOŚCI				
	23	Miesiące	1	2	3	4
	24	Pojazd_1	39 000,00	290 000,00	325 000,00	314 000,00
	25	Trasa_1	20 000,00	10 000,00	20 000,00	20 000,00
	26	Trasa_2	3 000,00	45 000,00	75 000,00	75 000,00
	27	Trasa_3	4 000,00	100 000,00	160 000,00	200 000,00
	28	Trasa_4	2 000,00	35 000,00	55 000,00	4 000,00
	29	Trasa_5	10 000,00	100 000,00	15 000,00	15 000,00
	30	Pojazd_2	29 000,00	210 000,00	224 500,00	229 000,00
	36	Razem	68 000,00	790 000,00	874 500,00	857 000,00
	37	Udział %	1%	7%	8%	
	38					

Rys. 2. Rozwiązanie nr 2

Źródło: fragment okna programu MS Excel.

dów (patrz: wiersze od 25 do 29 na rysunku 2). Ta zamiana ma na celu zmniejszenie ilości formuł wprowadzanych w pojedynczej komórce, co powinno ułatwić analizę tych formuł. Łatwiejsze powinno też być eliminowanie potencjalnych błędów.

Specyficzność wykorzystania funkcji SUMA w omawianym przypadku polega na wpisaniu kryterium sumowania wartości w postaci zakresów komórek spełniających warunek zgodności z daną trasą i grupą pojazdów oraz zastosowaniu formuły macierzowej tej funkcji. Postać i działanie omawianej funkcji przedstawiono w tabeli 5.

Tabela 5. Funkcja SUMA

Adres formuły	Formuła
B25	{=SUMA((\$A25&\$A\$24=\$D\$3:\$D\$12)*\$C\$3:\$C\$12)*B16}
Kroki	Wyniki działania szacowania funkcji
\$A25&\$A\$24	Trasa_1Pojazd_1
\$A25&\$A\$24=\$D\$3:\$D\$12	{Prawda/Falsz/Falsz/Falsz/Falsz/Falsz/Falsz/Falsz/Falsz}
(\$A25&\$A\$24=\$D\$3:\$D\$12)*\$C\$3:\$C\$12	{11/0/0/0/0/0/0/0/0}
SUMA((\$A25&\$A\$24=\$D\$3:\$D\$12)*\$C\$3:\$C\$12)	11
=SUMA(...	1 100

Źródło: opracowanie własne.

Jak przedstawiono w tabeli 5, działanie funkcji SUMA w takiej postaci jest bardzo podobne do działania funkcji SUMA.ILOCZYNÓW. Na początku zostało ustalone miejsce występowania w macierzy określonej trasy i grupy pojazdów poprzez zwrócenie wartości *Prawda* w określonym miejscu. Następnie wartości *Prawda* i *Falsz* są konwertowane na jedyne i zera oraz mnożone przez drugi argument, określający zakres komórek z podanymi cenami transportu. W wyniku tego otrzymujemy sumę tylko jednej komórki oznaczającej poszukiwaną wartość ceny dla konkretnej trasy i grupy pojazdów.

Wprowadzenie funkcji SUMA w postaci funkcji macierzowej rozszerza jej możliwości w znaczący sposób. Z funkcji stosowanej wyłącznie do sumowania wartości zamienia się w funkcję wyszukiwania na podstawie wielu kryteriów. Zastosowanie tej funkcji w postaci macierzowej wymaga jedynie użycia skrótu klawiaturowego Ctrl + Shift + Enter po zakończeniu edycji komórki z formułą. Tak przygotowaną funkcję należy wprowadzić do pozostałych wierszy przeznaczonych do obliczeń w ramach wszystkich tras.

5. Specyficzne zastosowanie funkcji SUMA i WYSZUKAJ.PIONOWO

Kolejną modyfikacją rozwiązania omawianego przykładu jest łączne zastosowanie dwu funkcji arkusza kalkulacyjnego, tj. SUMA i WYSZUKAJ.PIONOWO, w postaci macierzowej. To rozwiązanie polega na wprowadzeniu omawianych tu funkcji w dwie komórki równocześnie. Poprawne wprowadzenie formuły wymaga wykonania następujących kroków:

Krok 1. Zaznaczyć dwie przylegające komórki, jedna pod drugą.

Krok 2. Wprowadzić formułę, np. za pomocą paska formuły.

Krok 3. Po zakończeniu edycji wcisnąć skrót klawiaturowy Ctrl + Shift + Enter.

Na rysunku 3 przedstawiono strukturę arkusza wykorzystywanego przy prezentacji rozwiązania w tym podpunkcie. Arkusz ten różni się od poprzedniego tym, że pozostawiono tylko dwa dodatkowe wiersze (wiersze 24 i 26 na rysunku 3) przeznaczone na wprowadzenie formuł macierzowych. W tabeli 6 pokazano zastosowaną formułę oraz opisano jej działanie.

	A	B	C	D	E
1	CENNIK				
2	Trasa	Rodzaj obsługujących pojazdów	Index	Cena	
3	Trasa_1	Pojazd_1	Trasa_1Pojazd_1	200	
4	Trasa_1	Pojazd_2	Trasa_1Pojazd_2	150	
5	Trasa_2	Pojazd_1	Trasa_2Pojazd_1	300	
6	Trasa_2	Pojazd_2	Trasa_2Pojazd_2	200	
7	Trasa_3	Pojazd_1	Trasa_3Pojazd_1	400	
8	Trasa_3	Pojazd_2	Trasa_3Pojazd_2	300	
9	Trasa_4	Pojazd_1	Trasa_4Pojazd_1	100	
10	Trasa_4	Pojazd_2	Trasa_4Pojazd_2	50	
11	Trasa_5	Pojazd_1	Trasa_5Pojazd_1	500	
12	Trasa_5	Pojazd_2	Trasa_5Pojazd_2	400	
13					
14	HARMONOGRAM WYJAZDÓW				
15	Miesiące	1	2	3	4
16	Trasa_1	100	50	100	
17	Trasa_2	10	150	250	
18	Trasa_3	10	250	400	
19	Trasa_4	20	350	550	
20	Trasa_5	20	200	30	
21					
22	SUMA WARTOŚCI				
23	Pojazd_1	39000,00	290000,00	325000,00	31400,00
24		39000,00	290000,00	325000,00	31400,00
25	Pojazd_2	29000,00	210000,00	224500,00	22900,00
26		29000,00	210000,00	224500,00	22900,00
27	Razem	68 000,00	420 000,00	449 000,00	458 000,00
28					

Rys. 3. Rozwiązanie nr 3

Źródło: fragment okna programu MS Excel.

Porównując działanie funkcji macierzowej wprowadzanej do pojedynczej komórki i do zakresu komórek, uwidacznia się wyraźna różnica w ich działaniu. W przypadku funkcji wprowadzanej do jednej komórki przetwarzane są wszystkie dane zapisane jako argument funkcji. Przy wyświetlaniu wyniku jednakże pojawia się wartość odpowiadająca tylko pierwszym argumentom wprowadzanym właśnie jako macierz. W przypadku funkcji wprowadzanej jako zakres uwzględnia ona również w obliczeniach wszystkie komórki wskazane jako argumenty. Jednak wynik tej formuły jest wpisywany do wszystkich docelowych komórek i zawiera sumę po-

Tabela 6. Formuła SUMA i WYSZUKAJ.PIONOWO – wprowadzona do zakresu komórek

Adres formuły	Formuła
B23:B24	{=SUMA(WYSZUKAJ.PIONOWO(\$A\$16:\$A\$20&\$A23;\$C\$3:\$D\$12;2;0)*B\$16:B\$20)}
Kroki	Wyniki działania szacowania funkcji
\$A\$16:\$A\$20&\$A23	Trasa_1Pojazd_1
WYSZUKAJ.PIONOWO(\$A23&\$A\$16:\$A\$20;\$C\$3:\$D\$12;2;0)	11
WYSZUKAJ.PIONOWO(\$A\$16:\$A\$20&\$A23;\$C\$3:\$D\$12;2;0)*B\$16:B\$20	{1100/120/130/280/300}
=SUMA(...	1 930

Źródło: opracowanie własne.

szczególnych wyników obliczanych tak jak w przypadku formuły wprowadzanej do pojedynczej komórki. Różnice te można prześledzić, sprawdzając kolejne kroki szacowania formuły z tabel 6 i 7.

Tabela 7. Formuła SUMA i WYSZUKAJ.PIONOWO – wprowadzona do pojedynczej komórki

Adres formuły	Formuła
B23	{=SUMA(WYSZUKAJ.PIONOWO(\$A\$16:\$A\$20&\$A23;\$C\$3:\$D\$12;2;0)*B\$16:B\$20)}
Kroki	Wyniki działania szacowania funkcji
\$A\$16:\$A\$20&\$A23	Trasa_1Pojazd_1
WYSZUKAJ.PIONOWO(\$A23&\$A\$16:\$A\$20;\$C\$3:\$D\$12;2;0)	11
WYSZUKAJ.PIONOWO(\$A\$16:\$A\$20&\$A23;\$C\$3:\$D\$12;2;0)*B\$16:B\$20	{1100/120/130/280/300}
=SUMA(...	1 760

Źródło: opracowanie własne.

Jak wynika z analizy tabel 6 i 7 poprawne działanie omawianych formuł wymaga zastosowania wariantu z dwoma komórkami. Rozwiązanie to charakteryzuje się pewną dezynwolturą, ale właśnie taka składnia zapewnia poprawne rozwiązanie przy minimalnej komplikacji formuł.

6. Zmiana układu danych źródłowych i specyficzne zastosowanie funkcji SUMA

W tym podpunkcie zostanie omówione rozwiązanie polegające na użyciu funkcji SUMA w postaci macierzowej oraz na modyfikacji układu danych w tabeli *Cennik*. Zmiana tej tabeli wiąże się z prostym przekształceniem układu bazodanowego (ceny znajdują się w pojedynczej kolumnie) na układ tabelaryczny (ceny poszczególnych rodzajów środków transportowych znajdują się w oddzielnych kolumnach). Przekształcenie danych źródłowych wiąże się oczywiście z dodatkowymi operacjami obliczeniowymi, ale w rezultacie tego zabiegu otrzymuje się bardzo prostą i klarowną formułę rozwiązującą omawiany problem. Przekształcenia tabeli *Cennik* najłatwiej można dokonać za pomocą tabeli przestawnej (patrz rys. 4).

	A	B	C	D
22				
23	Suma z Cena	Etykiety kolumn ▾		
24	Etykiety wierszy ▾	Pojazd_1	Pojazd_2	
25	Trasa_1	200	150	
26	Trasa_2	300	200	
27	Trasa_3	400	300	
28	Trasa_4	100	50	
29	Trasa_5	500	400	
30				
31				

Rys. 4. Zamiana układu danych źródłowych z cenami za pomocą tabeli przestawnej

Źródło: fragment okna programu MS Excel.

Po przygotowaniu danych źródłowych do komórek z rozwiązaniem należy wprowadzić formułę przedstawioną w tabeli 8 w postaci macierzowej.

Tabela 8. Formuła SUMA

Adres formuły	Formuła
B18	{=SUMA(\$B\$3:\$B\$7*B\$11:B\$15)}
Kroki	Wyniki działania szacowania funkcji
\$B\$3:\$B\$7*B\$11:B\$15	{20000/3000/4000/2000/10000}
=SUMA(...	39 000,00

Źródło: opracowanie własne.

Jak pokazano w tabeli 8, użyta funkcja wykonuje obliczenia w taki sposób, że mnoży poszczególne argumenty podanych zakresów i następnie oblicza ich sumę. Jest to odpowiednik funkcji SUMA.ILOCZYNÓW, który oczywiście w tym przypadku również można zastosować.

Stosując to rozwiązanie, można zauważyć pewną niedogodność w kopiowaniu funkcji SUMA do wierszy w tabeli *Suma wartości*. Wiąże się to z koniecznością indywidualnego ustawiania adresu zakresu komórek z cenami używanych pojazdów w każdym wierszu. Ten mankament można zlikwidować poprzez zastosowanie definicji nazw dla zakresów komórek. W tym celu należy zaznaczyć dane wraz z nagłówkami w tabeli *Cennik* (na rysunku 4) i wybrać opcję z menu górnego Wstaw/Nazwa/Utwórz, a następnie zaznaczyć opcję Górny wiersz. W omawianym przypadku zostaną utworzone dwie nazwy *Pojazd_1* i *Pojazd_2*, które będzie można zastosować w formułach obliczających wartość użytkownika pojazdów. Aby jednak w pełni zautomatyzować wprowadzanie tych formuł, należy posłużyć się dodatkowo funkcją ADR.POŚR, która pozwoli na zamianę tekstu znajdującego się w boczku tabeli *Suma wartości* na adres rozpoznawalny przez funkcję SUMA. Zastosowanie tej modyfikacji pozwala na szybkie kopiowanie formuł poprzez wykorzystanie np. uchwytu przeciągania komórki. Wygląd zastosowanej formuły przedstawiono w tabeli 9.

Tabela 9. Funkcja SUMA i ADR.POŚR

Adres formuły	Formuła
B18	{=SUMA(ADR.POŚR(\$A18)*B\$11:B\$15)}
Kroki	Wyniki działania szacowania funkcji
\$A18	„Pojazd_1”
ADR.POŚR(\$A18)	{200/300/400/400/500}
\$B\$3:\$B\$7*B\$11:B\$15	{20000/30000/40000/20000/10000}
=SUMA(...	39 000,00

Źródło: opracowanie własne.

Prezentowane w tym podpunkcie rozwiązanie należy ocenić bardzo wysoko ze względu na następujące zalety:

- bardzo prosta i przejrzysta formuła,
- możliwość stosowania wariantu z funkcją SUMA.ILOCZYNÓW w przypadku chęci uniknięcia wprowadzania formuł macierzowych,
- łatwe rozszerzanie zakresów formuły obliczeniowej w przypadku dodawania nowych tras lub nowych grup pojazdów.

Oczywiście, za wadę tego rozwiązania można uznać konieczność konwersji układu danych źródłowych z bazodanowego na tabelaryczny, jednakże zastosowanie tabel przestawnych w tym przypadku zupełnie niweluje to utrudnienie.

7. Zmiana układu danych źródłowych i zastosowanie tabeli przestawnej

Ostatnim omawianym sposobem rozwiązania tytułowego problemu jest zastosowanie kolejnej modyfikacji danych źródłowych, z tym że modyfikacji zostaną podane dane dotyczące harmonogramów wyjazdów. Ta modyfikacja wydaje się naturalna ze względu na możliwości prowadzenia w MS Excelu ewidencji wyjazdów pojazdów i tego typu ewidencja może równocześnie stać się źródłem dla obliczeń przychodów związanych z użytkowaniem pojazdów. Układ tabeli harmonogramu wyjazdów po modyfikacji został pokazany na rysunku 5.

	A	B	C	D	E	F	G
1	CENNIK						
2	Trasa	Rodzaj obsługujących pojazdów	Index	Cena			
3	Trasa_1	Pojazd_1	Trasa_1Pojazd	200			
4	Trasa_1	Pojazd_2	Trasa_1Pojazd	150			
5	Trasa_2	Pojazd_1	Trasa_2Pojazd	300			
6	Trasa_2	Pojazd_2	Trasa_2Pojazd	200			
7	Trasa_3	Pojazd_1	Trasa_3Pojazd	400			
8	Trasa_3	Pojazd_2	Trasa_3Pojazd	300			
9	Trasa_4	Pojazd_1	Trasa_4Pojazd	100			
10	Trasa_4	Pojazd_2	Trasa_4Pojazd	50			
11	Trasa_5	Pojazd_1	Trasa_5Pojazd	500			
12	Trasa_5	Pojazd_2	Trasa_5Pojazd	400			
13							
14	HARMONOGRAM WYJAZDÓW						
15	Trasa	Miesiąc	Liczba wyjazdów	Pojazd_1	Pojazd_2	Wartość pojazd_1	Wartość pojazd_2
16	Trasa_1	1	100	200	150	20000	15000
17	Trasa_2	1	10	300	200	3000	2000
18	Trasa_3	1	10	400	300	4000	3000
19	Trasa_4	1	20	100	50	2000	1000
20	Trasa_5	1	20	500	400	10000	8000
21	Trasa_1	2	50	200	150	10000	7500
22	Trasa_2	2	150	300	200	45000	30000
23	Trasa_3	2	250	400	300	100000	75000
24	Trasa_4	2	350	100	50	35000	17500
25	Trasa_5	2	200	500	400	100000	80000
26	Trasa_1	3	100	200	150	20000	15000

Rys. 5. Zamiana układu danych źródłowych w tabeli *Harmonogram wyjazdów*

Źródło: fragment okna programu MS Excel.

Jak wynika z rysunku 5, w tabeli *Harmonogram wyjazdów* został zmieniony układ danych z tabelarycznego na bazodanowy oraz dodano kolumny: *Pojazd_1*, *Pojazd_2*, *Wartość pojazd_1* i *Wartość pojazd_2*. Taka struktura danych pozwala na rejestrację wyjazdów w poszczególnych miesiącach z dokładnością co do dnia. W tym wariancie rozwiązania nie będzie miała znaczenia kolejność wpisywania po-

szczególnych tras ani nawet kolejność wpisów do tabeli. Wiąże się to z wykorzystaniem tabeli przestawnej do generowania wyników ukazujących wartości generowane przez poszczególne rodzaje pojazdów. Na rysunku 5 zaznaczono żółtym kolorem kolumny posiadające atrybut do edycji. Pozostałe kolumny zawierają formuły wykonujące pomocnicze obliczenia. W kolumnach *Pojazd_1*, *Pojazd_2*, znajduje się funkcja WYSZUKAJ.PIONOWO zwracająca cenę odpowiadającą danej trasie i grupie pojazdów. W tabeli 10 pokazano działanie tej funkcji w przykładowej komórce.


Tabela 10. Funkcja WYSZUKAJ.PIONOWO

Adres formuły	Formuła
D16 ⇒ rys. 6	=WYSZUKAJ.PIONOWO(\$A16&D\$15;\$C\$2:\$D\$12;2;0)
Kroki	Wyniki działania szacowania funkcji
\$A16&D\$15	„Trasa_1Pojazd_1”
=WYSZ...	200

Źródło: opracowanie własne.

Natomiast kolumny *Wartość pojazd_1* i *Wartość pojazd_2* zawierają iloczyn ceny obliczonej na podstawie funkcji z tabeli 10 oraz liczby wyjazdów, znajdującej się w trzeciej kolumnie tabeli *Harmonogram wyjazdów*.

Utworzenie tabeli przestawnej jest ostatnim etapem tego rozwiązania. Po uruchomieniu kreatora *Raport tabeli przestawnej i wykresu przestawnego* należy podać jako źródło danych zakres komórek z tabeli *Harmonogram wyjazdów*. W układzie



Rys. 6. Definicja układu tabeli przestawnej

Źródło: fragment programu MS Excel.

tabeli przestawnej jako KOLUMNY definiuje się pole *Miesiące*, a do obszaru DANE przypisuje się pola *Wartość pojazd_1* i *Wartość pojazd_2*. Sposób definicji tabeli przestawnej przedstawiono na rysunku 6.

W przypadku prowadzenia ewidencji danych o wyjazdach nie wiadomo, jaki zakres komórek należy zaznaczyć w źródle tabeli przestawnej. W tym przypadku można się posłużyć kolejną nazwą zakresu, z tym że nazwa nie będzie się odwoływać do konkretnego zakresu danych, tylko zostanie przypisana do funkcji PRZESUNIĘCIE, której zadaniem będzie określanie na bieżąco aktualnego zakresu danych dla tabeli przestawnej. PRZESUNIĘCIE jest kolejną funkcją arkusza, która ma specyficzne działanie. Specyfika ta polega na tym, że wynikiem jej działania nie jest wartość, lecz zakres komórek. Argumenty tej funkcji określają miejsce tego zakresu względem dowolnej komórki w arkuszu oraz jego wielkość (liczba wierszy i kolumn). W omawianym przypadku, jako argument wyznaczający liczbę wierszy szukanego zakresu należy posłużyć się funkcją ILE.NIEPUSTYCH, która będzie np. sprawdzała, ile komórek zostało zapisanych w kolumnie *Ilość wierszy*. Postać tej funkcji przedstawiono wraz z jej argumentami w tabeli 11.

Tabela 11. Funkcja PRZESUNIĘCIE

Adres formuły	Formuła
formułę należy wpisać do pola Odwołuje się do okienka Definicji nazw	=PRZESUNIĘCIE(Arkusz1!\$A\$15;0;0;ILE.NIEPUSTYCH(Arkusz1!\$C\$15:\$C\$500);7)
Argumenty	Opis
Arkusz1!\$A\$15	jest to adres komórki bazowej, względem której następuje przesunięcie szukanego zakresu
0	liczba wierszy, o które należy przesunąć lewą górną komórkę szukanego zakresu w stosunku do komórki bazowej
0	liczba kolumn, o które należy przesunąć lewą górną komórkę szukanego zakresu w stosunku do komórki bazowej
ILE.NIEPUSTYCH(Arkusz1!\$C\$15:\$C\$500);	liczba wierszy w szukanym zakresie, obliczane na nowo za każdym przeliczeniem arkusza dzięki zastosowaniu funkcji ILE.NIEPUSTYCH
7	liczba kolumn w szukanym zakresie (w tym przypadku jest to liczba kolumn w tabeli Harmonogram wyjazdów)

Źródło: opracowanie własne.

Funkcję przedstawioną w tabeli 11 należy wpisać w odpowiednie miejsce okna Definicji nazw i dodać do dowolnej nazwy opisującej *Harmonogram wyjazdów*. Jeżeli tę nazwę zastosuje się jako źródło danych dla tabeli przestawnej, to po każdym rozszerzeniu o kolejne wyjazdy, trasy czy dni w tabeli z ewidencją wyjazdów oraz odświeżeniu tabeli przestawnej uwzględnione zostaną nowe wpisy.

8. Zakończenie

W niniejszym opracowaniu przedstawiono szereg możliwości arkusza kalkulacyjnego MS Excel, które zostały określone przez autora jako specyficzne. Należy jednak zdawać sobie sprawę, że pewne aspekty przedstawionych funkcji lub narzędzi są stosunkowo rzadko stosowane przez użytkowników tego arkusza, zaliczających się do średniozaawansowanych z powodu braku świadomości istnienia tego typu możliwości. W związku z czym głównym celem tego opracowania jest poszerzenie umiejętności właśnie tych użytkowników.

Na zakończenie należy podsumować pokazane warianty w poszczególnych podpunktach. Pierwszy wariant z funkcją SUMA.ILOCZYNÓW charakteryzuje się bardzo długą i skomplikowaną formułą, co można uznać za dość uciążliwą wadę, zwłaszcza w przypadku dochodzenia nowych tras czy grup pojazdów. Drugi wariant z funkcją SUMA w postaci macierzowej również daje rozbudowaną formułę, ale poprzez zastosowanie dodatkowych wierszy oraz złączenie dwóch kryteriów wyszukiwania w jedną kolumnę okazuje się rozwiązaniem dość czytelnym i dającym się łatwo modyfikować. Wariant 3 z funkcją SUMA i WYSZUKAJ.PIONOWO w postaci macierzowej daje najbardziej skondensowaną formułę, co jest dużą zaletą w porównaniu z poprzednimi rozwiązaniami. Wariant ten nie jest oczywiście pozbawiony wady, którą niewątpliwie stanowi konieczność wprowadzania formuły do dwóch komórek równocześnie.

Warianty 4 i 5 wiążą się z założeniem, że istnieje możliwość dokonywania zmian w strukturze danych źródłowych. W wariantcie 4 dokonano modyfikacji danych cennikowych, w wariantcie 5 zmiana dotyczyła tabeli z ewidencją liczby wyjazdów. Właśnie te dwa ostatnie rozwiązania wydają się – zdaniem autora – najbardziej efektywne, co wskazuje na duży związek z wcześniejszym określeniem wyglądu danych źródłowych, zanim zostaną podjęte decyzje co do sposobu przetwarzania danych. Jeżeli chcemy dopasować rozwiązanie do już zebranych danych, można się spodziewać różnych niedogodności, które właśnie zostały wskazane.

Mimo wskazówek autora dotyczących poszczególnych rozwiązań omawianego problemu ważna jest dla użytkownika wiedza o wielowariantowości każdego zagadnienia związanego z przetwarzaniem danych w arkuszu kalkulacyjnym. Dobór wariantu zależy od wiedzy, doświadczenia oraz szczegółowych ograniczeń w konkretnych przypadkach.

ADVANCED COMPUTATIONAL FORMULAE IN THE SOLUTION OF PROBLEM OF CALCULATION OF PRICE OF LOGISTICAL ATTENDANCE

Summary: The article presents the description of problems of employment of imputed sheet in the controlling of logistics. Information support in this range is a very important element of information system. In a situation when the system uses mainly imputed sheets it is important to employ the advanced functions and formulae which help the effective utilization of the instrument. The author includes array formulae to these functions .